```
   20 ca_create path,'pma2'
...
```

Now `f` has been defined to have `pma2` symmetry and running `ca_create` with this definition of `mkrandf` creates examples with that symmetry. We use `winbd` and `fullness` for the frieze groups as previously defined. We finish the section by generating the finite part of all the remaining frieze symmetry groups.

```
   ]id=: =i. 3                      the identity matrix
1 0 0
0 1 0
0 0 1

   ]in1=: _1 0 0,0 _1 0,:0 0 1      central inversion (2-fold rotation)
_1  0 0
 0 _1 0
 0  0 1

   ]mr1=: 1 0 0,0 _1 0,:0 0 1       reflection across the x-axis
1  0 0
0 _1 0
0  0 1

   $p111=: Prods^:_~ ,:id           no non-translational symmetry
1 3 3

   $p112=: Prods^:_~ ,:in1          2-fold rotation around origin
2 3 3

   $pm11=: Prods^:_~ ,:mr0          mirrors perpendicular to the
2 3 3                               translation

   $p1m1=: Prods^:_~ ,:mr1          mirrors parallel to the translation
2 3 3

   $p1a1=: Prods^:_~ ,:gl1          glide reflection
2 3 3

   $pmm2=: Prods^:_~ mr0,:mr1       parallel and perpendicular mirrors
4 3 3
```

Now using any of these arrays as an argument to `mkrandfrz` in place of `pma2` will result in attractors with those symmetries. Figure 9.7.3 illustrates chaotic attractors with the other five frieze symmetries created using this approach.

9.8 Experiment: Crystallographic Symmetry on a Square Lattice

We now turn to investigating symmetric chaotic attractors that are periodic in two different directions in the plane. The symmetry groups of patterns that have two periodic directions are called the planar crystallographic groups. These are also known as the wallpaper groups. There are analogous groups for three and higher dimensions which are also known as crystallographic groups. Remarkably, the 230 crystallographic groups in three-space were enumerated before the 17 planar groups were enumerated. It turns out that 12 of the 17 planar crystallographic groups can be constructed on a square lattice although many of those could also be exhibited on a more general lattice. The other five planar symmetry groups involve 3-fold or 6-fold rotations and require a hexagonal lattice which is discussed in Section 9.9. In this section, we develop the tools we need to explore 12 crystallographic groups on a square lattice.

First we can create functions which are periodic modulo one in both coordinates by taking a Fourier series in one coordinate with coefficients that are Fourier series in the other coordinate. This is known as a double Fourier series. The double Fourier series mod 1 is constructed with the adverb `DF`. We assume *chaotica.ijs* has been loaded.

```
   DF
1 : '1 |((m +/ .*{:)+/ . *{.)@:(four"0)@(2p1&*)'

   mkrandDF=: 2 : '((m*_1 1) randunif 2 5 5) DF'

   mkrandf=: 0.14 mkrandDF

   f=: ''mkrandf

   f 0.1 0.2
0.930178 0.886152

   f 0.1 0.2+3 _2                      f is periodic in both coordinates
0.930178 0.886152
```

Notice that the function `f` created in this way is periodic mod 1. Also, the choice of scale for the randomly selected parameters, `0.14`, deserves some comment. As the function types become more complicated, choosing a scale so that the Ljapunov exponents frequently fall into the desired interval becomes delicate. We can run some experiments to try to determine effective choices of scale as follows. First we define a function `L_randf` that gets the Ljapunov exponent estimates for a function created by `mkrandf`; the argument to `L_randf` is not used except that the experiment is repeated for each element of the argument array since `L_randf` has rank zero. Running it five or more times for various choices of scale for the random parameters gives us experience with the resulting Ljapunov exponents.

```
   L_randf
3 : 'f Lexp (f=:0 mkrandf)^:(200) 0.3 0.2'"0

   mkrandf=: 0.2 mkrandDF          scale of 0.2 gives Ljapunov exponents that are too big

   L_randf i.5
1.21124 0.569112
1.24655  0.47808
1.19987 0.592844
1.24528 0.439024
1.02635 0.215802

   mkrandf=: 0.05 mkrandDF         scale of 0.05 gives Ljapunov exponents that are too small

    L_randf i.5
_0.298404  _0.647176
  _0.33733  _0.327776
_0.0640147 _0.0529328
 _0.108673    _1.33091
 _0.275764  _0.296532

   mkrandf=: 0.14 mkrandDF         scale of 0.14 gives a mix of Ljapunov exponents

   L_randf i.5
_0.104416 _0.0933238
 0.789656  _0.111571
 0.507049  _0.285267
  0.76056 _0.0318023
 0.851544   0.129105
```

As in the case for the frieze groups, we need to create finite groups of generators by taking products of appropriate generating symmetries and then reducing any translation modulo one. The functions with the desired equivariance properties are the identity added to the sum of conjugates modulo one. This is somewhat simpler than the frieze case since the identity function and the reduction mod one occur in



Figure 9.8.1 Chaotic Attractors with Six Crystallographic Symmetries



pmg



pgg



cmm



p4



p4m



p4g

Figure 9.8.2 Chaotic Attractors with Six Other Crystallographic Symmetries

both coordinates. Indeed, except for those differences, `crycon` is identical to `frzcon`. We also need to create appropriate functions for defining the window bounds (the unit square) and for determining the fullness of an image (whether all rows and columns are visited, adjoined to the number of visited rows and columns).

```
   crycon
2 : 0
mp=.(+/ . *)"1 2
1 | ] + }:@(+/)@(] mp (%.n)"_)@:((,1:)@u@}:"1)@(] mp n"_)@(,1:) f.
)
```

```
   winsq                          window is the unit square
0 0 1 1"_
```

```
   fullsq
3 : '(([:*./#=+/),+/)(+./,.+./"1)*y'
```

```
   winbd=: winsq
```

```
   fullness=: fullsq
```

The first crystallographic group we investigate is *p2* which is generated by a 2-fold rotation about the origin.

```
   ]i11=: _1 0 0,0 _1 0 ,:0 0 1           two-fold rotation
_1  0 0
 0 _1 0
 0  0 1
```

```
   $p2=: Prods^:_~ ,:i11
2 3 3
```

```
   mkrandcry
2 : ' m mkrandDF 0 crycon n'
```

```
   mkrandf=: 1 : '0.15 mkrandcry p2'
```

Running `ca_create` with the above functions gives rise to chaotic attractors; one of which was run high resolution is shown as the *p2* attractor in Figure 9.8.1. Notice the translations in the vertical and horizontal directions and the 2-fold rotations. Ignoring translation by the period, how many different 2-fold rotations can you identify in that image?

The second planar crystallographic group we will discuss is denoted *cm*. It can be generated by a parallel mirror and glide reflection; alternately, following The International Tables of Crystallography, we can generate it with a reflection and a translation by `0.5 0.5`.

```
   ]i10=: _1 0 0 ,0 1 0,:0 0 1            a reflection
_1 0 0
 0 1 0
 0 0 1
```

```
   ]h11=: 1 0 0,0 1 0,:0.5 0.5 1          a translation
  1   0 0
  0   1 0
0.5 0.5 1
```

```
   $cm=: Prods^:_~ i10,:h11
4 3 3
```

```
   mkrandf=: 1 : '0.06 mkrandcry cm'
```

Running `ca_create` with that adverb gives rise to chaotic attractors such as the one shown in the *cm* example in Figure 9.8.1. Notice the vertical mirrors and the glide reflections between the mirrors. Ignoring translation by the periods, how many different reflection lines do you see? How many different glide reflections do you see? Look at the array `cm`. It contains 4 items. Two of the items are the generators and another is the identity matrix. Geometrically describe the fourth symmetry in `cm`.

In order to tile several copies of the images produced by `ca_create`, `ca_hr` or `ca_hr_add` together to highlight the periodic symmetries, we use the function `tile_ image` that we used in Section 9.7.

```
   path=: 'c:\temp\cm\'

   20 ca_create path,'cm_a'
...

   3 3 &tile_image&> 'cm_a*.png' files_in path
```

Figures 9.8.1 and 9.8.2 give illustrations of all the planar crystallographic symmetry groups that can be realized on the square lattice. You should be able, by observation and experimentation, to identify symmetries in each of those illustrations and thereby create examples of that type. Select one symmetry group from each figure, except *p2* or *cm*, and create chaotic attractors with those symmetries.

9.9 Crystallographic Symmetry on a Hexagonal Lattice

The crystallographic groups that involve the hexagonal lattice require some special attention. These groups do not require glide reflections in their generator lists, so we are actually able to use simpler constructions for creating our random functions with the desired symmetry. However, since the lattice isn't square, we need to reduce modulo the lattice more carefully. The target parallelogram is equivalent to a rectangle and then that rectangle needs to be offset in order to tile with the appropriate symmetry.

Before we consider the hexagonal symmetry groups, we will pause to review the "under" conjunction. The function `f` under `g` is denoted `f&.g` and this results in the inverse (or obverse) of `g` atop `f` composed with `g`. Many familiar functions are related by this pattern. For example, multiplication is addition under the logarithm.

```
   3 * 4
12

   3 +&.^. 4
12
```

As a second illustration of the use of under, consider computing the Euclidean length of a vector: we want the square root of the sum of the squares of the entries. Since the square root and squaring are inverse, we can use under. However, we need to be careful with the rank, we want to use `f&.:g` which gives the infinite rank version of under. Once we have an expression for determining the Euclidean length of a vector, a hook gives us a utility for computing unit vectors.

```
   +/&.*: 1 _2 2                  sums absolute values of each item;
1 2 2                             not what we want

   +/&.:*: 1 _2 2                 root of the sums of squares
3

   unit=: % +/&.:*:               unit vector function

   unit 1 2 2                     unit vector in 1 2 2 direction
0.333333 0.666667 0.666667

   unit 1 1
0.707107 0.707107
```

Now we create a hexagonal lattice. The vectors `v1` and `v2` defined below are $2\pi/3$ radians apart.